
CHAPTER 22

Using Query of Queries
A query that retrieves data from a record set is called a Query of Queries. After you 
generate a record set, you can interact with its results as if they were database tables by 
using Query of Queries. This chapter describes the benefits and procedures for this 
feature.

About record sets
Query of Queries is based on manipulating the record set, which you can create using the 
cfquery tag and other ways.

When you execute a database query, ColdFusion retrieves the data in a record set. In 
addition to presenting record set data to the user, you can manipulate this record set to 
improve your application’s performance.

Because a record set contains rows (records) and columns (fields), you can think of it as a 
virtual database table, or as a spreadsheet. For example, the cfpop tag retrieves a record set 
in which each row is a message and each column is a message component, such as To, 
From, and Subject.

Referencing queries as objects
You can reference ColdFusion queries as objects by assigning a query to a variable, as 
follows:

<cfquery name = "query01"
  datasource = "myDNS"
  SELECT * FROM CUSTOMERS
</cfquery>
...
<cfset query02 = query01> 

The query is not copied; both names point to the same record set data. Therefore, if you 
make changes to the table referenced in query01, the original query and the query object 
called query02 both reflect these changes. If you perform a copy with an array, the array 
is copied.

Creating a record set
You can perform a Query of Queries on any ColdFusion tag or function that generates a 
record set, including the following:
• cfcollection
• cfdirectory
• cfftp
• cfhttp
• cfindex
• cfldap

• cfmail
• cfpop
• cfprocresult

• cfquery (against a database or against another Query of Queries)
• cfsearch
• cfstoredproc

• cfwddx
• the queryNew( ); query function 

Creating a record set with a function
In addition to creating a record set by using a cfquery or other CFML tags, you can 
create it with the queryNew() function.

To create a record set with the queryNew() function:

1 Create a ColdFusion page with the following content:
<html>
<head>
<title>The queryNew function</title>
</head>

<body>
<h2>QueryNew Example</h2>

<!--- create a query ---><cfset qInstruments = queryNew("name, instrument, 
years_playing")>

<!--- add rows --->
<cfset newrow  = queryaddrow(qInstruments, 3)>

<!--- set values in cells --->
<cfset temp = querysetcell(qInstruments, "name", "Thor", 1)>
<cfset temp = querysetcell(qInstruments, "instrument", "hammer", 1)>
<cfset temp = querysetcell(qInstruments, "years_playing", "1000", 1)>

<cfset temp = querysetcell(qInstruments, "name", "Bjorn", 2)>
<cfset temp = querysetcell(qInstruments, "instrument", "sitar", 2)>
<cfset temp = querysetcell(qInstruments, "years_playing", "24", 2)>

<cfset temp = querysetcell(qInstruments, "name", "Raoul", 3)>
<cfset temp = querysetcell(qInstruments, "instrument", "flute", 3)>
<cfset temp = querysetcell(qInstruments, "years_playing", "12", 3)>

<!--- output the query --->
<cfoutput query="qInstruments">

<pre>#name##instrument# #years_playing#</pre>
</cfoutput>

<h3>Individual record retrieval:</h3>
<cfoutput>
<p>#qInstruments.name[2]# has played #qInstruments.instrument[2]# for

#qInstruments.years_playing[2]# years.
</cfoutput>

</body>
</html>

2 Save the page as queryNew.cfm in the myapps directory under the web_root directory.

3 In your browser, enter the following URL to display the query results:

http://127.0.0.1/myapps/queryNew.cfm

The following figure shows how the output appears:

Note: When you create a record set, you can store in it complex objects, such as arrays and 
structures. However, you cannot use Query Of Queries on a record set that contains complex 
objects. For more information on Query of Queries, see “About Query of Queries” on page 
465.

About Query of Queries
After you have created a record set with a tag or function, you can query the record set in 
several dependent queries. A query that retrieves data from a record set is called a Query 
of Queries. A typical use of a Query of Queries is to retrieve an entire table into memory 
with one query, and then access the table data (the record set) with subsequent sorting or 
filtering queries. In essence, you query the record set as if it were a database table.

Note: Because you can generate a record set in ways other than using the cfquery tag, the 
term In Memory Query is sometimes used instead of Query of Queries.

Benefits of Query of Queries
Performing a Query of Queries has many benefits, including the following:
• If you need to access the same tables multiple times, you greatly reduce access time, 

because the data is already in memory (in the record set). 
A Query of Queries is ideal for tables of 5,000 to 50,000 rows, and is limited only by 
the memory of the ColdFusion Server host machine.

• You can perform joins and union operations on results from different data sources. 
For example, you can perform a union operation on queries from different databases 
to eliminate duplicates for a mailing list.

• You can efficiently manipulate cached query results in different ways. You can query a 
database once, and then use the results to generate several different summary tables.
For example, if you need to summarize the total salary by department, by skill, and 
by job, you can make one query to the database and use its results in three separate 
queries to generate the summaries.

• You can obtain drill-down, master-detail information for which you do not access the 
database for the details.
For example, you can select information about departments and employees in a 
query, and cache the results. You can then display the employees’ names. When users 
select an employee, the application displays the employee’s details by selecting 
information from the cached query, without accessing the database.

Performing a Query of Queries
There are four steps to perform a Query of Queries. 

To perform a Query of Queries:

1 Generate a record set.

You can write a normal query using a tag or function that creates a record set. This is 
sometimes called a master query. For more information, see “Creating a record set” 
on page 462.

2 Write a detail query—a query that specifies dbtype="query" in its cfquery tag.

3 In the detail query, write a SQL statement that retrieves the relevant records. Specify 
the names of one or more existing queries as the table names in your SQL code. Do 
not specify a datasource attribute.

4 If the database content does not change rapidly, use the cachedwithin attribute of the 
master query to cache the query results between page requests. This way, ColdFusion 
accesses the database on the first page request, and does not query the database again 
until the specified time expires. You must use the CreateTimeSpan function to specify 
the cachedwithin attribute value (in days, hours, minutes, seconds format).

The detail query generates a new query results set, identified by the value of the name 
attribute of the detail query. The following example illustrates the use of a master query 
and a single detail query that extracts information from the master.

To use the results of a query in a query:

1 Create a ColdFusion page with the following content:
<body>
<h1>Employee List</h1>
<!--- LastNameSearch (normally generated interactively) --->
<cfset LastNameSearch="Doe">

<!--- Master Query --->
<cfquery datasource="CompanyInfo" name="master"

cachedwithin=#CreateTimeSpan(0,1,0,0)#>
SELECT * from Employee

</cfquery>

<!--- Detail Query (dbtype=query, no data source) --->
<cfquery dbtype="query" name="detail"> 

SELECT Emp_ID, FirstName, LastName
FROM master
WHERE LastName=<cfqueryparam value="#LastNameSearch#" 

cfsqltype="cf_sql_char" maxLength="20"></cfquery>

<!--- output the detail query results --->
<p>Output using a query of query:</p>
<cfoutput query=detail>

#Emp_ID#: #FirstName# #LastName#<br>
</cfoutput>
<br>

<p>Columns in the master query:</p>
<cfoutput>

#master.columnlist#<br>
</cfoutput>
<br>

<p>Columns in the detail query:</p>
<cfoutput>

#detail.columnlist#<br>
</cfoutput>
</body>

2 Save the page as query_of_query.cfm in the myapps directory under the web_root.

3 In your browser, enter the following URL to display the queryresults:

http://127.0.0.1/myapps/query_of_query.cfm

The following figure shows how the output appears:

Reviewing the code

The master query retrieves the entire Employee table from the CompanyInfo data source 
(the CompanyInfo database). The detail query selects only the three columns to display 
for employees with the specified last name. The following table describes the code and its 
function:

Code Description

cfset LastNameSearch="Doe" Sets the last name to use in the detail 
query. In a complete application, this 
information comes from user interaction.

<cfquery datasource="CompanyInfo" name="master"
cachedwithin=#CreateTimeSpan(0,1,0,0)#>
SELECT * from Employee

</cfquery>

Queries the CompanyInfo data source and 
selects all data in the Employees table. 
Caches the query data between requests 
to this page, and does not query the 
database if the cached data is less than an 
hour old.

<cfquery dbtype="query" name="detail">
SELECT Emp_ID, FirstName, LastName
FROM master
WHERE LastName=<cfqueryparam
value=”#LastNameSearch#" 
cfsqltype=”cf_sql_char” 
maxLength="20"></cfquery>

Uses the master query as the source of 
the data in a new query, named detail. This 
new query selects only entries that match 
the last name specified by the 
LastNameSearch variable. The query also 
selects only three columns of data: 
employee ID, first name, and last name. 
The query uses the cfqueryparam tag to 
prevent passing erroneous or harmful 
code.

Displaying record set data incrementally

If your database is large, you can limit the number of rows displayed at one time. The 
following example shows how to use the currentRow query variable of a Query of Queries 
to do this. For more information on query variables, see “Getting information about 
query results” on page 441.

To display record set data incrementally:

1 Create a ColdFusion page with the following content:
<html>
<head>
<title>QoQ with incremental row return</title>
</head>

<body>
<h3>QoQ with incremental row return</h3>
<!--- define startrow and maxrows to facilitate 'next N' style browsing --->
<cfparam name = "MaxRows" default = "5">
<cfparam name = "StartRow" default = "1">

<!--- master query: retrieve all info from Employee table --->
<cfquery name = "GetSals" datasource = "CompanyInfo">

SELECT * FROM  Employee
ORDER BY LastName

</cfquery>

<!--- detail query: select 3 fields from the master query --->
<cfquery name = "GetSals2" dbtype = "query">

SELECT   FirstName, LastName, Salary
FROM     GetSals
ORDER BY LastName

</cfquery>

<!--- build table to display output --->
<table cellpadding = 1 cellspacing = 1>

<tr>
  <td bgcolor = f0f0f0>
  <b><i>&nbsp;</i></b>
  </td>

<cfoutput query=detail>
#Emp_ID#: #FirstName# #LastName# <br>

</cfoutput>

Uses the detail query to display the list of 
employee IDs, first names, and last 
names.

<cfoutput>
#master.columnlist#<br>

</cfoutput>

Lists all the columns returned by the 
master query.

<cfoutput>
#detail.columnlist#<br>

</cfoutput>

Lists all the columns returned by the detail 
query.

Code Description

  <td bgcolor = f0f0f0>
  <b><i>FirstName</i></b>
  </td>

  <td bgcolor = f0f0f0>
  <b><i>LastName</i></b>
  </td>

  <td bgcolor = f0f0f0>
  <b><i>Salary</i></b>
  </td>
</tr>

<!--- Output the query and define the startrow and maxrows
   parameters. Use the query variable currentRow to
   keep track of the row you are displaying. --->
<cfoutput query = "GetSals2" startrow = "#StartRow#" maxrows = "#MaxRows#">
<tr>
  <td valign = top bgcolor = ffffed>

  <b>#GetSals2.currentRow#</b>
  </td>

  <td valign = top>
  <font size = "-1">#FirstName#</font>

  </td>

  <td valign = top>
  <font size = "-1">#LastName#</font>

  </td>

  <td valign = top>
  <font size = "-1">#Salary#</font>

  </td>
</tr>
</cfoutput>

<!--- If the total number of records is less than or equal to 
the total number of rows, provide a link to the same page, with the 
StartRow value incremented by MaxRows (5, in this example) --->
<tr>
  <td colspan = 4>
  <cfif (startrow + maxrows) lte getsals2.recordcount>
    <a href="qoq_next_row.cfm?startrow=<cfoutput>#Evaluate(StartRow +

MaxRows)#</cfoutput>">See next <cfoutput>#MaxRows#</cfoutput> 
    rows</a>
  </cfif>  
  </td>
</tr>

</table>
</body>
</html>

2 Save the page as qoq_next_row.cfm in the myapps directory under the web_root.

3 In your web browser, enter the following URL to display the query results:

http://127.0.0.1/myapps/qoq_next_row.cfm
About Query of Queries 469



The following figure shows how the output appears:

Using the cfdump tag with query results

As you debug your CFML code, you can use the cfdump tag to quickly present the value 
of your query. This tag has the following format:

<cfdump var="#query_name#">

For more information on the cfdump tag, see CFML Reference.

Using Query of Queries with non-SQL record sets

A Query of Queries can operate on any CFML tag or function that returns a record set; 
you are not limited to operating on cfquery results. You can perform queries on non-SQL 
record sets, such as a cfdirectory tag, Verity searches, a cfldap tag, and so on. 

The following example shows how a Query of Queries interacts with the record set of a 
Verity search. This example assumes that you have a valid Verity collection, called bbb, 
which contains documents with a target word, film, or its variants (films, filmed, 
filming). Change the name of the collection and the search criteria to as appropriate for 
your Verity collection. For more information on Verity, see Chapter 24, “Building a 
Search Interface” on page 521.

To use Query of Queries with a Verity record set:

1 Create a ColdFusion page with the following content:
<html>
<head>
<title>QoQ and Verity</title>
</head>

<body>
<!--- master query: retrieve all documents from the bbb collection 
that contain ’film’ (or its stemmed variants); change values for 
collection and criteria as needed for your Verity collection --->

<cfsearch name = "quick"
collection="bbb" 
type = "simple"
criteria="film">

<h3>Master query dump:</h3>
<cfdump var="#quick#">

<!--- detail query: retrieve from the master query only those 
documents with a score greater than a criterion (here, 
0.7743) --->
<cfquery name="qoq" dbtype="query">

SELECT * from quick
WHERE quick.score > 0.7743

</cfquery>

<h3>Detail query dump:</h3>
<cfdump var="#qoq#">

</body>
</html>

2 Save the page as qoq_verity.cfm in the myapps directory under the web_root.

3 In your web browser, enter the following URL to display the query results:

http://127.0.0.1/myapps/qoq_verity.cfm

The following figure shows how the output appears:

Note: This figure shows a collapsed master query output and an expanded detail query 
output. Click an output to expand or collapse it.

The first cfdump tag shows the master query, which retrieves all records. The second 
cfdump shows the Query of Queries results.

Tip: Adjust the score criterion of the detail query to reflect the contents of your 
collection.

The next example shows how a Query of Queries combines record sets from a 
cfdirectory tag, which is limited to retrieval of one file type per use.

To use Query of Queries to combine record sets:

1 Create a ColdFusion page with the following content:
<html>
<head>
<title>Images Folder</title>
</head>

<body>
<h2>Image Retrieval with QoQ</h2> 
<!--- set the images directory --->
<cfset dir = ("C:\pix\")>

<!--- retrieve all GIFs --->
<cfdirectory name="GetGIF"
  action="list"
  directory="#dir#"
  filter="*.gif">

<!--- retrieve all JPGs --->
<cfdirectory name="GetJPG"
  action="list"
  directory="#dir#"
  filter="*.jpg">
  
<!--- join the queries with a UNION in a QoQ (cfdirectory 

automatically returns the directory name as "Name") --->
<cfquery dbtype="query" name="GetBoth">
  SELECT * FROM GetGIF
  UNION 
  SELECT * FROM GetJPG
  ORDER BY Name
</cfquery>

<!--- display output in a linked, ordered list --->
<cfoutput>
  <p>The <strong>#dir#</strong> directory contains #GetBoth.RecordCount#

images:<br>
  <ol>
    <cfloop query="GetBoth">
      <li><a href="../images/#Name#">#GetBoth.Name#</a><br>
    </cfloop>  
  </ol>
</cfoutput>

</body>
</html>

2 Save the page as qoq_cfdirectory.cfm in the myapps directory under the web_root.

3 In your web browser, enter the following URL to display the query results:

http://127.0.0.1/myapps/qoq_cfdirectory.cfm

The following figure shows how the output appears:

Query of Queries user guide
The following sections discuss Query of Queries functionality. If you know SQL or have 
interacted with databases, you might be familiar with some of these features.

Using dot notation
ColdFusion supports using dot notation in table names.

Example

If a structure named A contains a field named B, which contains a table named Products, 
you can refer to the table with dot notation, as follows:

SELECT tape_ID, length
FROM A.B.Products;

Using joins
A join operation uses a single SELECT statement to return a result set from multiple 
tables. The re are two main types of JOIN operations:
• INNER JOIN includes in the result set only records that are present in both tables
• OUTER JOIN includes in the result set all records in one of the tables.

ColdFusion does not support OUTER JOINs, nor does it support the INNER JOIN 
syntax, as the following example shows:

SELECT Dog_ID, Breed_ID,
FROM Dogs INNER JOIN Breed
ON Dogs.Dog_ID = Breed.Dog_ID;

ColdFusion supports INNER JOINs between two tables, as the following example 
shows. This operation is the most common type of join.

SELECT Dog_ID, Breed_ID
FROM Dogs, Breed
WHERE Dogs.Dog_ID = Breed.Dog_ID;

Using unions
The UNION operator lets you combine the results of two or more SELECT expressions 
into a single record set. The original tables must have the same number of columns, and 
corresponding columns must be UNION-compatible data types. Columns are 
UNION-compatible data types if they meet one of the following conditions:
• The same data type; for example, both Tinyint
• Both Numeric; for example, Tinyint, Smallint, Integer, Bigint, Double, Float, Real, 

Decimal, or Numeric
• Both Characters; for example, Char, Varchar, or LongVarchar
• Both Dates; for example, Time, TimeStamp, or Date

Note: Query Of Queries does not support ODBC-formatted dates and times.

Syntax

select_expression = select_expression UNION [ALL] select_expression

Example

This example uses the following tables:

To combine Table1 and Table2, use a UNION statement, as follows:

SELECT * FROM Table1 
UNION 
SELECT * FROM Table2 

The UNION statement produces the following result (UNION) table:

Using aliases for column names

The column names of a UNION table are the column names in the result set of the first 
SELECT statement in the UNION operation; ColdFusion ignores the column names in 
the other SELECT statement. To change the column names of the result table, you can 
use an alias, as follows:

Select Type as SportType, Name as SportName from Table1
UNION
Select * from Table2

Table1

Type(int) Name(varchar)

1 Tennis

2 Baseball

3 Football

Table2

ID(int) Sport(varchar)

3 Football

4 Volleyball

5 PingPong

Result table

Type(int) Name(varchar)

1 Tennis

2 Baseball

3 Football

4 Volleyball

5 PingPong

Duplicate rows and multiple tables

By default, the UNION operator removes duplicate rows from the result table. If you use 
the keyword ALL, then duplicates are included.

You can combine an unlimited number of tables using the UNION operator, for 
example: 

Select * from Table1
UNION 
Select * from Table2
UNION 
Select * from Table3
... 

Parentheses and evaluation order

By default, the Query of Queries SQL engine evaluates a statement containing UNION 
operators from left to right. You can use parentheses to change the order of evaluation. 
For example, the following two statements are different: 

/* First statement. */ 
SELECT * FROM TableA
UNION ALL 
(SELECT * FROM TableB 
UNION 
SELECT * FROM TableC
) 

/* Second statement. */ 
(SELECT * FROM TableA 
UNION ALL 
SELECT * FROM TableB 
) 
UNION 
SELECT * FROM TableC 

In the first statement, there are no duplicates in the union between TableB and TableC. 
Then, in the union between that set and TableA, the ALL keyword includes the 
duplicates. In the second statement, duplicates are included in the union between TableA 
and TableB but are eliminated in the subsequent union with TableC. The ALL keyword 
has no effect on the final result of this expression. 

Using other keywords with UNION

When you perform a UNION, the individual SELECT statements cannot have their 
own ORDER BY or COMPUTE clauses. You can only have one ORDER BY or 
COMPUTE clause after the last SELECT statement; this clause is applied to the final, 
combined result set. You can only specify GROUP BY and HAVING expressions in the 
individual SELECT statements. 

Using conditional operators
ColdFusion lets you use the following conditional operators in your SQL statements:
• Test
• Null
• Comparison
• Between
• IN
• LIKE

Test conditional

This conditional tests whether a Boolean expression is true, false, or unknown. 

Syntax

cond_test ::= expression [IS [NOT] {TRUE | FALSE | UNKNOWN} ]

Example

SELECT _isValid FROM Chemicals
WHERE _isValid IS true;

Null conditional

This conditional tests whether an expression is null.

Syntax

null_cond ::= expression IS [NOT] NULL 

Example

SELECT bloodVal FROM Standards
WHERE bloodVal IS NOT null;

Comparison conditional

This conditional lets you compare an expression against another expression of the same 
data type (Numeric, String, Date, or Boolean). You can use it to selectively retrieve only 
the relevant rows of a record set.

Syntax

comparison_cond ::= expression [> | >= | <> | != | < | <=] expression 

Example

The following example uses a comparison conditional to retrieve only those dogs whose 
IQ is at least 150:

SELECT dog_name, dog_IQ
FROM Dogs
WHERE dog_IQ >= 150;

Between conditional

This conditional lets you compare an expression against another expression. You can use 
it to selectively retrieve only the relevant rows of a record set. Like the comparison 
conditional, the BETWEEN conditional makes a comparison; however, the between 
conditional makes a comparison against a range of values. Therefore, its syntax requires 
two values, which are inclusive, a minimum and a maximum. You must separate these 
values with the AND keyword.

Syntax

between_cond ::= expression [NOT] BETWEEN expression AND expression 

Example

The following example uses a BETWEEN conditional to retrieve only those dogs whose 
IQ is between 150 and 165, inclusive:

SELECT dog_name, dog_IQ
FROM Dogs
WHERE dog_IQ BETWEEN 150 AND 165;

IN conditional

This conditional lets you specify a comma-delimited list of conditions to match. It is 
similar in function to the OR conditional. In addition to being more legible when 
working with long lists, the IN conditional can contain another SELECT statement.

Syntax

in_cond ::= expression [NOT] IN (expression_list) 

Example

The following example uses the IN conditional to retrieve only those dogs who were born 
at either Ken’s Kennels or Barb’s Breeders:

SELECT dog_name, dog_IQ, Kennel_ID
FROM Dogs
WHERE kennel_ID IN ('Kens','Barbs');

LIKE conditional

This conditional lets you perform wildcard searches, in which you compare your data to 
search patterns. This strategy differs from other conditionals, such as BETWEEN or IN, 
because the LIKE conditional compares your data to a value that is partially unknown.

Syntax

like_cond ::= left_string_exp [NOT] LIKE right_string_exp [ESCAPE escape_char] 

The left_string_exp can be either a constant string, or a column reference to a string 
column. The right_string_exp can be either a column reference to a string column, or a 
search pattern. A search pattern is a search condition that consists of literal text and at 
least one wildcard character. A wildcard character is a special character that represents an 
unknown part of a search pattern, and is interpreted as follows: 
• The underscore (_) represents any single character. 
• The percent sign (%) represents zero or more characters.
• Square brackets ([ ]) represents any character in the range.
• Square brackets with a caret [^] represent any character not in the range.
• All other characters represent themselves. 

Note: Earlier versions of ColdFusion do not support bracketed ranges.

Examples

The following example uses the LIKE conditional to retrieve only those dogs of the breed 
Terrier, whether the dog is a Boston Terrier, Jack Russell Terrier, Scottish Terrier, and so 
on:

SELECT dog_name, dog_IQ, breed
FROM Dogs
WHERE breed LIKE '%Terrier';

The following examples are select statements that use bracketed ranges:

SELECT lname FROM Suspects WHERE lname LIKE 'A[^c]%';
SELECT lname FROM Suspects WHERE lname LIKE '[a-m]%';
SELECT lname FROM Suspects WHERE lname LIKE '%[]';
SELECT lname FROM Suspects WHERE lname LIKE 'A[%]%';
SELECT lname FROM Suspects WHERE lname LIKE 'A[^c-f]%';

Case sensitivity

ColdFusion supports two string functions, UPPER() and LOWER(), which you can use to 
achieve case-insensitive matching.

Examples

The following example matches only 'Sylvester':

SELECT dog_name
FROM Dogs
WHERE dog_name LIKE 'Sylvester';

The following example is not case-sensitive; it uses the LOWER() function to match 
'Sylvester', 'sylvester', 'SYLVESTER', and so on:

SELECT dog_name
FROM Dogs
WHERE LOWER(dog_name) LIKE 'Sylvester';

Escaping wildcards

You can specify your own escape character using the conditional ESCAPE clause.

Example

The following example uses the ESCAPE clause to enable a search for a literal percent 
sign (%), which ColdFusion normally interprets as a wildcard character:

SELECT emp_discount
FROM Benefits
WHERE emp_discount LIKE '10\%' 
ESCAPE '\';

Using aggregate functions
Aggregate functions operate on a set of data and return a single value. Use these functions 
for retrieving summary information from a table, as opposed to retrieving an entire table 
and then operating on the record set of the entire table.

Consider using aggregate functions to perform the following operations:
• To display the average of a column
• To count the number of rows for a column
• To find the earliest date in a column

Since not every relational database management system (RDBMS) supports all aggregate 
functions, refer to your database’s documentation. The following table lists the aggregate 
functions that ColdFusion supports:

Note: For more information, see CFML Reference.

Syntax

aggregate_func ::= <COUNT>(* | column_name) | AVG | SUM | MIN | MAX)
([ALL | DISTINCT] numeric_exp)

Example

The following example uses the AVG() function to retrieve the average IQ of all terriers:

SELECT dog_name, AVG(dog_IQ) AS avg_IQ
FROM Dogs
WHERE breed LIKE '%Terrier';

Function Description

AVG() Returns the average (mean) for a column.

COUNT() Returns the number of rows in a column.

MAX() Returns the largest value of a column.

MIN() Returns the lowest value of a column.

SUM() Returns the sum of values of a column.

Arbitrary expressions in aggregate functions

ColdFusion supports aggregate functions of any arbitrary expression, as follows:

SELECT lorange, count(lorange+hirange) 
FROM roysched
GROUP BY lorange;

Aggregate functions in arbitrary expressions

ColdFusion supports mathematical expressions that include aggregate functions, as 
follows:

SELECT MIN(lorange) + MAX(hirange)
FROM roysched
GROUP BY lorange;

Using group by and having expressions
ColdFusion supports the use of any arbitrary arithmetic expression, as long as it is 
referenced by an alias.

Examples

The following code is correct:

SELECT (lorange + hirange)/2 AS midrange,
COUNT(*)
FROM roysched
GROUP BY midrange;

The following code is correct:

SELECT (lorange+hirange)/2 AS x, 
COUNT(*)
FROM roysched GROUP BY x 
HAVING x > 10000;

The following code is not supported in Query of Queries:

SELECT (lorange + hirange)/2 AS midrange,
COUNT(*)
FROM roysched
GROUP BY (lorange + hirange)/2;

Using ORDER BY clauses
ColdFusion supports the ORDER BY clause to sort. Make sure that it is the last clause in 
your SELECT statement. You can sort by multiple columns, by relative column position, 
by nonselected columns. You can specify a descending sort direction with the DESC 
keyword (by default, most RDBMS sorts are ascending, which makes the ASC keyword 
unnecessary).

Syntax

order_by_column ::= ( <IDENTIFIER> | <INTEGER_LITERAL> ) [<ASC> | <DESC>]

Examples

The following example shows a simple sort using an ORDER BY clause:

SELECT acetylcholine_levels, dopamine_levels
FROM results
ORDER BY dopamine_levels

The following example shows a more complex sort; results are first sorted by ascending 
levels of dopamine, then by descending levels of acetylcholine. The ASC keyword is 
unnecessary, and is used only for legibility.

SELECT acetylcholine_levels, dopamine_levels
FROM results
ORDER BY 2 ASC, 1 DESC

Using aliases
ColdFusion supports the use of database column aliases. An alias is an alternate name for 
a database field or value. ColdFusion lets you reuse an alias in the same SQL statement.

One way to create an alias is to concatenate (append) two or more columns to generate a 
value. For example, you can concatenate a first name and a last name to create the value 
fullname. Because the new value does not exist in a database, you refer to it by its alias. 
The AS keyword assigns the alias in the SELECT statement.

Examples

ColdFusion supports alias substitutions in the ORDER BY, GROUP BY, and HAVING 
clauses. 

Note: ColdFusion does not support aliases for table names.

SELECT FirstName + ' ' + LastName AS fullname
from Employee;

The following examples rely on these two master queries:

<cfquery name="employee" datasource="2pubs">
SELECT * FROM employee

</cfquery>

<cfquery name="roysched" datasource="2pubs">
SELECT * FROM roysched

</cfquery>

To generate output for the following examples, use the cfdump tag. For example, use 
<cfdump var="#order_by#"> for the following ORDER BY example.

ORDER BY example

<cfquery name="order_by" dbtype="query">
SELECT (job_id + job_lvl)/2 AS job_value
FROM employee 
ORDER BY job_value

</cfquery>

GROUP BY example

<cfquery name="group_by" dbtype="query">
SELECT lorange+hirange AS x, count(hirange) 
FROM roysched 
GROUP BY x

</cfquery>

HAVING example

<cfquery name="having" dbtype="query">
SELECT (lorange+hirange)/2 AS x, 
COUNT(*)
FROM roysched GROUP BY x 
HAVING x > 10000

</cfquery>

Handling null values
ColdFusion uses Boolean logic to handle conditional expressions. Proper handling of 
NULL values requires the use of ternary logic. The IS [NOT] NULL clause works correctly 
in ColdFusion. However the following expressions do not work properly when the 
column breed is NULL:

WHERE (breed > 'A')
WHERE NOT (breed > 'A') 

The correct behavior should not include NULL breed columns in the result set of either 
expression. To avoid this limitation, you can add an explicit rule to the conditionals and 
rewrite them in the following forms:

WHERE breed IS NOT NULL AND (breed > 'A')
WHERE breed IS NOT NULL AND not (breed > 'A')

Escaping reserved keywords
ColdFusion has a list of reserved keywords, which are typically part of the SQL language 
and are not normally used for names of columns or tables. To escape a reserved keyword 
for a column name or table name, enclose it in brackets.

Caution: Earlier versions of ColdFusion let you use some reserved keywords without 
escaping them.

Examples

ColdFusion supports the following SELECT statement examples:

SELECT [from] FROM parts;
SELECT [group].firstname FROM [group];
SELECT [group].[from] FROM [group];

ColdFusion does not support nested escapes, such as in the following example:

SELECT [[from]] FROM T;

The following table lists ColdFusion reserved keywords: 

ABSOLUTE ACTION ADD ALL ALLOCATE

ALTER AND ANY ARE AS

ASC ASSERTION AT AUTHORIZATION AVG

BEGIN BETWEEN BIT BIT_LENGTH BOTH

BY CASCADE CASCADED CASE CAST

CATALOG CHAR CHARACTER CHARACTER_LENGTH CHAR_LENGTH

CHECK CLOSE COALESCE COLLATE COLLATION

COLUMN COMMIT CONNECT CONNECTION CONSTRAINT

CONSTRAINTS CONTINUE CONVERT CORRESPONDING COUNT

CREATE CROSS CURRENT CURRENT_DATE CURRENT_TIME

CURRENT_TIMESTAMP CURRENT_USER CURSOR DATE DAY

DEALLOCATE DEC DECIMAL DECLARE DEFAULT

DEFERRABLE DEFERRED DELETE DESC DESCRIBE

DESCRIPTOR DIAGNOSTICS DISCONNECT DISTINCT DOMAIN

DOUBLE DROP ELSE END END-EXEC

ESCAPE EXCEPT EXCEPTION EXEC EXECUTE

EXISTS EXTERNAL EXTRACT FALSE FETCH

FIRST FLOAT FOR FOREIGN FOUND

FROM FULL GET GLOBAL GO

GOTO GRANT GROUP HAVING HOUR

IDENTITY IMMEDIATE IN INDICATOR INITIALLY

INNER INPUT INSENSITIVE INSERT INT

INTEGER INTERSECT INTERVAL INTO IS

ISOLATION JOIN KEY LANGUAGE LAST

LEADING LEFT LEVEL LIKE LOCAL

LOWER MATCH MAX MIN MINUTE

MODULE MONTH NAMES NATIONAL NATURAL

NCHAR NEXT NO NOT NULL

NULLIF NUMERIC OCTET_LENGTH OF ON

ONLY OPEN OPTION OR ORDER

OUTER OUTPUT OVERLAPS PAD PARTIAL

POSITION PRECISION PREPARE PRESERVE PRIMARY

PRIOR PRIVILEGES PROCEDURE PUBLIC READ

REAL REFERENCES RELATIVE RESTRICT REVOKE

RIGHT ROLLBACK ROWS SCHEMA SCROLL

SECOND SECTION SELECT SESSION SESSION_USER

SET SIZE SMALLINT SOME SPACE

SQL SQLCODE SQLERROR SQLSTATE SUBSTRING

SUM SYSTEM_USER TABLE TEMPORARY THEN

TIME TIMESTAMP TIMEZONE_HOUR TIMEZONE_MINUTE TO

TRAILING TRANSACTION TRANSLATE TRANSLATION TRIM

TRUE UNION UNIQUE UNKNOWN UPDATE

UPPER USAGE USER USING VALUE

VALUES VARCHAR VARYING VIEW WHEN

WHENEVER WHERE WITH WORK WRITE

YEAR ZONE

BNF for Query of Queries
The Backus Naur Form (BNF) is a formal notation to describe programming syntax. The 
following is the BNF for Query of Queries:

Input ::= select_statement 

select_statement ::= select_expression ( <ORDER> <BY> order_by_list )? 

select_expression ::= ( <OPENPAREN> select_expression <CLOSEPAREN> |
select_specification ) ( <UNION> ( <ALL> )? select_expression )? 

select_specification ::= <SELECT> ( <ALL> | <DISTINCT> )? select_list <FROM>
from_table_list ( <WHERE> cond_exp )? ( <GROUP> <BY> group_by_list )? 
( <HAVING> cond_exp )? 

order_by_list ::= order_by_column ( <COMMA> order_by_column )* 

order_by_column ::= ( <IDENTIFIER> | <INTEGER_LITERAL> ) ( <ASC> | <DESC> )? 

group_by_list ::= column_ref ( <COMMA> column_ref )* 

from_table_list ::= <IDENTIFIER> ( <COMMA> <IDENTIFIER> )* 

select_list ::= select_column ( <COMMA> select_column )* 

select_column ::= <ASTERISK> 
 | <IDENTIFIER> <DOT> ( <ASTERISK> | <IDENTIFIER> ( alias )? ) 
 | expression ( alias )? 

alias ::= ( <AS> )? <IDENTIFIER> 

cond_exp ::= cond_term ( <OR> cond_exp )? 

cond_term ::= cond_factor ( <AND> cond_term )? 

cond_factor ::= ( <NOT> )? cond_test 

cond_test ::= cond_primary ( <IS> ( <NOT> )? ( <TRUE> | <FALSE> | <UNKNOWN> ) )? 

cond_primary ::= simple_cond 
 | <OPENPAREN> cond_exp <CLOSEPAREN> 

simple_cond ::= like_cond 
 | null_cond 
 | between_cond 
 | in_cond 
 | comparison_cond 

null_cond ::= row_constructor <IS> ( <NOT> )? <NULL> 

comparison_cond ::= row_constructor comparison_operator row_constructor 

between_cond ::= row_constructor ( <NOT> )? <BETWEEN> row_constructor 
<AND> row_constructor 

in_cond ::= row_constructor ( <NOT> )? <IN> <OPENPAREN> ( expression_list )
<CLOSEPAREN> 

row_constructor ::= expression 

comparison_operator ::= <LESSEQUAL> 
 | <GREATEREQUAL> 
 | <NOTEQUAL> 
 | <NOTEQUAL2> 
 | <EQUAL> 
 | <LESS> 
 | <GREATER> 

like_cond ::= string_exp ( <NOT> )? <LIKE> string_exp 

expression_list ::= expression ( <COMMA> expression )? 

expression ::= <STRING_LITERAL> 
 | <OPENPAREN> <STRING_LITERAL> <CLOSEPAREN> 
 | numeric_exp 

numeric_exp ::= numeric_term ( ( <PLUS> | <MINUS> ) numeric_exp )? 

numeric_term ::= numeric_factor ( ( <ASTERISK> | <SLASH> ) numeric_term )? 

numeric_factor ::= ( <PLUS> | <MINUS> )? numeric_primary 

numeric_primary ::= <INTEGER_LITERAL> 
 | <FLOATING_POINT_LITERAL> 
 | aggregate_func 
 | column_ref 
 | <OPENPAREN> numeric_exp <CLOSEPAREN> 

aggregate_func ::= <COUNT> <OPENPAREN> count_param <CLOSEPAREN> 
 | ( <AVG> | <SUM> | <MIN> | <MAX> ) <OPENPAREN> ( <ALL> | <DISTINCT> )?

numeric_exp <CLOSEPAREN> 

count_param ::= <ASTERISK> 
 | ( <ALL> | <DISTINCT> )? numeric_exp 

string_exp ::= <STRING_LITERAL> 
 | column_ref 
 | <OPENPAREN> string_exp <CLOSEPAREN> 

column_ref ::= <IDENTIFIER> ( <DOT> <IDENTIFIER> )?

